SageMath: Open source software for symbolic mathematics[1]

Jerry Brunner, Department of Statistical Sciences

[1] This slide show is an open-source document. See last slide for copyright information.

Symbolic calculation

- Lots of software can carry out numerical calculations, and so can `SageMath`.
- What makes `SageMath` special is that it can also do *symbolic* computation.
- That is, it is able to manipulate symbols as well as numbers.

- How much of the "math" we do is actually a set of clerical tasks?

Free and open source

The `SageMath` website is `http://www.sagemath.org`.

- There are commercial versions like
  - Mathematica (http://www.wolfram.com)
  - Maple (http://www.maplesoft.com)
- `SageMath` is not the only free alternative.
- It includes quite a few of the others.
- Python based.
- On your computer, or use `SageMathCloud`.

SageMath on your computer: A free download

- It's a big download.
- Native versions for Mac and linux.
- Windows version lives in a virtual linux machine.
- First download Oracle's VirtualBox.

Web browser interface

When you start up for the first time you will see something like this:

Click on New Worksheet.

Give it a name



Check Typeset



Click in the box.

Click in the box: evaluate appears



Type code in the box and click evaluate.

Click evaluate



Type more code in the next box.

Keep going



The Sage Notebook
Version 4.3
admin | Toggle | Home | Published | Log | Settings | Report a Problem | Help | Sign out
Tour1
last edited on April 26, 2012 11:41 AM by admin
Save
Save & quit
Discard & quit
File...
Action
Data...
sage
Typeset
Print
Worksheet
Edit
Text
Undo
Share
Publish
1+1
2
1 + 1.0
2.00000000000000
evaluate

Integer arithmetic



Same calculation with R

`SageMath`'s answer was 11652140094042840181048771286026412160

```
> prod(1:100)/(prod(1:60)*prod(1:30)*prod(1:10))

[1] 1.165214e+37
```

Charlotte's homework problem

$$\frac{4x}{4x^2 - 8x + 7} + \frac{3x}{4x^2 - 10x + 7} = 1$$

Note $16 \times 7 = 112$.

$$16x^4 - 100x^3 + 200x^2 - 175x + 49 = 0$$

Use SageMath

Solve $16x^4 - 100x^3 + 200x^2 - 175x + 49 = 0$

```
f = 16*x^4 - 100*x^3 + 200*x^2 - 175*x + 49
factor(f)
```

evaluate

$$\left(4\,x^{2} - 9\,x + 7\right)(2\,x - 1)(2\,x - 7)$$

Solve directly

$\frac{4x}{4x^2-8x+7} + \frac{3x}{4x^2-10x+7} = 1$

```
eq = 4*x/(4*x^2-8*x+7) + 3*x/(4*x^2-10*x+7) == 1
solve(eq,x) # Solve eq for x
```

evaluate

$$\left[x = \left(\frac{1}{2}\right), x = \left(\frac{7}{2}\right), x = -\frac{1}{8} i \sqrt{31} + \frac{9}{8}, x = \frac{1}{8} i \sqrt{31} + \frac{9}{8}\right]$$

Derivatives

Calculate $\frac{\partial^3}{\partial x^3}\left(\frac{e^{4x}}{1+e^{4x}}\right)$. This is something you could do by hand, but would you want to?

```
f(x) = exp(4*x)/(1+exp(4*x))
derivative(f(x),x,3) # Derivative with respect to x, 3 times
```

evaluate

$$\frac{64\,e^{(4\,x)}}{e^{(4\,x)}+1} - \frac{448\,e^{(8\,x)}}{\left(e^{(4\,x)}+1\right)^2} + \frac{768\,e^{(12\,x)}}{\left(e^{(4\,x)}+1\right)^3} - \frac{384\,e^{(16\,x)}}{\left(e^{(4\,x)}+1\right)^4}$$

```
factor(_) # Factor the preceding expression
```

evaluate

$$\frac{64\left(e^{(8\,x)}-4\,e^{(4\,x)}+1\right)e^{(4\,x)}}{\left(e^{(4\,x)}+1\right)^4}$$

Look at $f(x)$ again

```
f(x)
```

evaluate

$$\frac{e^{(4\,x)}}{e^{(4\,x)}+1}$$

It looks like a cumulative distribution function. Is it?

Check $\lim_{x\to-\infty} f(x)$ and $\lim_{x\to\infty} f(x)$

```
limit(f(x),x=-Infinity);limit(f(x),x=Infinity)
```

evaluate

0

1

Or,

```
show(limit(f(x),x=-Infinity))
show(limit(f(x),x=Infinity))
```

evaluate

0

1

The (single) derivative of $f(x)$ is a density.

Since $f(x)$ is a cumulative distribution function

```
derivative(f(x),x)
```

evaluate

$$4\,\frac{e^{(4\,x)}}{\left(e^{(4\,x)}+1\right)} - 4\,\frac{e^{(8\,x)}}{\left(e^{(4\,x)}+1\right)^{2}}$$

```
# Another way
f(x).derivative(x)
```

evaluate

$$\frac{4\,e^{(4\,x)}}{e^{(4\,x)}+1} - \frac{4\,e^{(8\,x)}}{\left(e^{(4\,x)}+1\right)^{2}}$$

Simplify and save the density

```
g(x) = factor(f(x).derivative(x)); g(x)
```

evaluate

$$\frac{4\,e^{(4\,x)}}{\left(e^{(4\,x)}+1\right)^{2}}$$

Plot $g(x) = \frac{4\,e^{(4\,x)}}{\left(e^{(4\,x)}+1\right)^2}$

```
plot(g(x),x,-5,5)
```

evaluate



Is $g(x)$ symmetric around zero?

If so, then expected value = median = 0.

```
f(0)
```

evaluate

$\frac{1}{2}$

Symmetry about zero means $g(x) = g(-x)$

```
g(x)-g(-x)
```

evaluate

$$\frac{4\,e^{(4\,x)}}{\left(e^{(4\,x)}+1\right)^2} - \frac{4\,e^{(-4\,x)}}{\left(e^{(-4\,x)}+1\right)^2}$$

```
factor(g(x)-g(-x))
```

evaluate

0

Is this right? Yes. To see it, just multiply numerator and denominator of $g(-x) = \frac{4\,e^{-4\,x}}{(e^{-4\,x}+1)^2}$ by e^{8x}, obtaining $g(x) = \frac{4\,e^{4\,x}}{(e^{4\,x}+1)^2}$.

Replace 4 with θ in $f(x) = \frac{e^{4\,x}}{e^{4\,x}+1}$

Need to declare any symbolic variable other than x.

```
var('theta')
F(x) = exp(theta*x)/(1+exp(theta*x)); F(x)
```

evaluate

$\frac{e^{(\theta x)}}{e^{(\theta x)}+1}$

Is $F(x) = \frac{e^{\theta\, x}}{e^{\theta\, x}+1}$ a distribution function?

```
limit(F(x),x=-Infinity)
```

evaluate

```
Traceback (click to the left of this block for traceback)
...
Is  theta  positive, negative, or zero?
```

- Good question!
- Actually, the question is asked by the excellent open-source calculus program `Maxima`, and `Sage` relays the question.
- In `Maxima`, you could answer the question interactively.
- In `SageMath`, click in the box and edit the code.

Assume $\theta > 0$

```
assume(theta>0)
F(x).limit(x=-oo); F(x).limit(x=oo)
```

evaluate

0

1

Differentiate to get the density.

```
# New f(x) will replace the old one.
f(x) = factor(F(x).derivative(x)); f(x)
```

evaluate

$$\frac{\theta e^{(\theta x)}}{\left(e^{(\theta x)}+1\right)^2}$$

$f(x)$ is symmetric about zero.

So the expected value must be zero.

```
factor(f(x)-f(-x)) # Checking symmetry
```

evaluate

0

```
# Expected value
integrate(x*f(x),x,-oo,oo)
```

evaluate

0

The variance emerges in terms of an obscure function called the `polylog`. The calculation will not be shown.

Add a location parameter

```
var('mu')
F(x) = exp(theta*(x-mu))/(1+exp(theta*(x-mu))); F(x)
```

evaluate

$$\frac{e^{(-(\mu-x)\theta)}}{e^{(-(\mu-x)\theta)}+1}$$

I can't control the order of variables in `SageMath` output. It looks alphabetical, with the `m` in `mu` coming before x.

The density is tasteless

As SageMath writes it

```
f(x) = factor( F(x).derivative(x) ); f(x)
```

evaluate

$$\frac{\theta e^{(\mu\theta+\theta x)}}{\left(e^{(\mu\theta)}+e^{(\theta x)}\right)^2}$$

By hand,

$$\begin{aligned} f(x) &= \frac{\theta e^{(\mu\theta+\theta x)}}{\left(e^{\mu\theta}+e^{\theta x}\right)^2} \cdot \frac{e^{-2\mu\theta}}{e^{-2\mu\theta}} \\ &= \cdots \\ &= \frac{\theta e^{\theta(x-\mu)}}{\left(e^{\theta(x-\mu)}+1\right)^2} \end{aligned}$$

Simulating from the distribution

- A good source of cute homework problems.
- Give students the density $f(x) = \frac{\theta e^{\theta(x+\mu)}}{\left(e^{\theta x}+e^{\theta\mu}\right)^2}$.
- μ can be estimated with $\overline{x}$ but it's not obvious.
- θ has to be estimated numerically.
- Need to generate some data from this distribution.

Simulation by the inverse CDF method

A well-known rule

- Let $F(x)$ be the CDF of a continuous random variable with $F \uparrow$ on its support, so that F^{-1} exists.
- If $U \sim U(0,1)$, then $X = F^{-1}(U) \sim F(x)$.
- In words, if U is a random variable with a uniform density on the interval $(0,1)$ and $F(x)$ is the cumulative distribution function of a continuous random variable, then if you transform U with the *inverse* of $F(x)$, the result is a random variable with distribution function $F(x)$.

Of course you could do this by hand, but . . .

```
# Inverse of cdf
var('X U')
solve(F(X)==U,X) # Solve F(X)=U for X
```

evaluate

$$\left[X = \frac{\mu\theta + \log\left(-\frac{U}{U-1}\right)}{\theta}\right]$$

It might be a bit better to write this as

$$X = \mu + \frac{1}{\theta}\log\left(\frac{U}{1-U}\right),$$

but what `Sage` gives us is quite nice.

Simulate data from $F(x)$ using R

$X = \mu + \frac{1}{\theta} \log\left(\frac{U}{1-U}\right)$

```
> n = 20; mu = -2; theta = 4
> U = runif(n)
> X = mu + log(U/(1-U))/theta; X

 [1] -1.994528 -2.455775 -2.389822 -2.996261 -1.477381 -2.422011 -1.855653
 [8] -2.855570 -2.358733 -1.712423 -2.075641 -1.908347 -2.018621 -2.019441
[15] -1.956178 -2.015682 -2.846583 -1.727180 -1.726458 -2.207717
```

Some symbolic variables do not need to be declared

```
pi
```

evaluate

π

Is that really the ratio of a circle's circumference to its diameter, or just the Greek letter?

```
cos(pi)
```

evaluate

-1

That's pretty promising. Evaluate it numerically.

```
n(pi) # Could also say pi.n()
```

evaluate

3.14159265358979

Try using pi in the normal distribution

```
# Normal density
var('mu, sigma')
assume(sigma>0)
f(x) = 1/(sigma*sqrt(2*pi)) * exp(-(x-mu)^2/(2*sigma^2)); f(x)
```

evaluate

$$\frac{\sqrt{2}e^{\left(-\frac{(\mu-x)^2}{2\,\sigma^2}\right)}}{2\,\sqrt{\pi}\sigma}$$

```
# Integrate the density
integrate(f(x),x,-oo,oo)
```

evaluate

1

Expected value and variance of the normal

Direct calculation

```
# E(X)
integrate(x*f(x),x,-oo,oo)
```

evaluate

μ

```
# E(X-mu)^2 is the variance
integrate((x-mu)^2*f(x),x,-oo,oo)
```

evaluate

σ^2

Moment-generating function

```
# Moment-generating function M(t) = E(e^{Xt})
var('t')
M(t) = integrate(exp(x*t)*f(x),x,-oo,oo); M(t)
```

evaluate

$$e^{\left(\frac{1}{2}\,\sigma^2 t^2 + \mu t\right)}$$

```
# Differentiate four times, set t=0 to get E(X^4)
derivative(M(t),t,4)(t=0)
```

evaluate

$$\mu^4 + 6\,\mu^2\sigma^2 + 3\,\sigma^4$$

Geometric distribution

A coin with $Pr\{\text{Head}\} = \theta$ is tossed repeatedly, and X is the number of tosses required to get the first head.

```
# Geometric
var('theta')
assume(0<theta); assume(theta<1)  # Note two statements
p(x) = theta*(1-theta)^(x-1); p(x)
p(x).sum(x,1,oo)                 # Sum the pmf
```

evaluate

1

Expected value and variance of the Geometric

```
(x*p(x)).sum(x,1,oo)                # Expected value
```

evaluate

$\frac{1}{\theta}$

```
((x-1/theta)^2*p(x)).sum(x,1,oo) # Variance
```

evaluate

$-\frac{\theta-1}{\theta^2}$

Linear algebra

Not even scratching the surface

- The algorithm that Sage uses for a particular task will depend on the *ring* (a concept from Algebra) to which the matrix belongs.
- When the contents of a matrix are symbols, the matrix belongs to the symbolic ring, abbreviated `SR`.
- As in Python, a matrix is a list of rows, and the rows are lists of matrix elements.

A symbolic matrix

```
var('alpha beta gamma delta')
A = matrix( SR, [[alpha, beta],[gamma, delta]] ); A
```

evaluate

$$\begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}$$

As in Python, index numbering begins with zero, not one.

```
A[0,1]
```

evaluate

β

Multiplication by a scalar does what you would hope

```
a=2
a*A
```

evaluate

$$\begin{pmatrix} 2\,\alpha & 2\,\beta \\ 2\,\gamma & 2\,\delta \end{pmatrix}$$

Matrix multiplication

```
var('x11 x12 x13 x21 x22 x23')
B = matrix(SR, [[x11, x12, x13], [x21, x22, x23]])
B
```

evaluate

$$\begin{pmatrix} x_{11} & x_{12} & x_{13} \\ x_{21} & x_{22} & x_{23} \end{pmatrix}$$

```
C = A*B; C
```

evaluate

$$\begin{pmatrix} \alpha x_{11} + \beta x_{21} & \alpha x_{12} + \beta x_{22} & \alpha x_{13} + \beta x_{23} \\ \gamma x_{11} + \delta x_{21} & \gamma x_{12} + \delta x_{22} & \gamma x_{13} + \delta x_{23} \end{pmatrix}$$

Of course the matrices must be the right size or `SageMath` raises an error.

Matrix transpose etc.

Using a notation that quickly becomes natural if it is not already

```
show(A)
A.transpose()
```

evaluate

$$\begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}$$

$$\begin{pmatrix} \alpha & \gamma \\ \beta & \delta \end{pmatrix}$$

Trace and determinant

$A = \begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}$

```
A.trace()
```

evaluate

$\alpha + \delta$

```
A.determinant()
```

evaluate

$\alpha\delta - \beta\gamma$

SageMath has a scrollbar

```
D = C.transpose() # C is 2x3, D is 3x2
E = (C*D).inverse()   # Inverse of C*D
factor(E) # E is HUGE! This is not as bad. Factor is a good way to simplify.
```

evaluate

$$\left(\begin{array}{l} \frac{\gamma^2 x_{11}^2+\gamma^2 x_{12}^2+\gamma^2 x_{13}^2+2\,\delta\gamma x_{11}x_{21}+\delta^2 x_{21}^2+2\,\delta\gamma x_{12}x_{22}+\delta^2 x_{22}^2+2\,\delta\gamma x_{13}x_{23}+\delta^2 x_{23}^2}{(x_{12}^2x_{21}^2+x_{13}^2x_{21}^2-2\,x_{11}x_{12}x_{21}x_{22}+x_{11}^2x_{22}^2+x_{13}^2x_{22}^2-2\,x_{11}x_{13}x_{21}x_{23}-2\,x_{12}x_{13}x_{22}x_{23}+x_{11}^2x_{23}^2+x_{12}^2x \cdots} \\ -\frac{\alpha\gamma x_{11}^2+\alpha\gamma x_{12}^2+\alpha\gamma x_{13}^2+\alpha\delta x_{11}x_{21}+\beta\gamma x_{11}x_{21}+\beta\delta x_{21}^2+\alpha\delta x_{12}x_{22}+\beta\gamma x_{12}x_{22}+\beta\delta x_{22}^2+\alpha\delta x_{13}x_{23}+\beta\gamma x_{1} \cdots}{(x_{12}^2x_{21}^2+x_{13}^2x_{21}^2-2\,x_{11}x_{12}x_{21}x_{22}+x_{11}^2x_{22}^2+x_{13}^2x_{22}^2-2\,x_{11}x_{13}x_{21}x_{23}-2\,x_{12}x_{13}x_{22}x_{23}+x_{11}^2x_{23}^2+x_{12}^2x \cdots} \end{array} \right.$$

A smaller example

Recall $\mathbf{A} = \begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}$

```
A.inverse() # Here is something we can look at without a scrollbar.
```

evaluate

$$\begin{pmatrix} \frac{1}{\alpha} + \frac{\beta\gamma}{\alpha^2\left(\delta - \frac{\beta\gamma}{\alpha}\right)} & -\frac{\beta}{\alpha\left(\delta - \frac{\beta\gamma}{\alpha}\right)} \\ -\frac{\gamma}{\alpha\left(\delta - \frac{\beta\gamma}{\alpha}\right)} & \frac{1}{\delta - \frac{\beta\gamma}{\alpha}} \end{pmatrix}$$

```
Ainverse = factor(_) # Factor the last expression.
Ainverse
```

evaluate

$$\begin{pmatrix} \frac{\delta}{\alpha\delta - \beta\gamma} & -\frac{\beta}{\alpha\delta - \beta\gamma} \\ -\frac{\gamma}{\alpha\delta - \beta\gamma} & \frac{\alpha}{\alpha\delta - \beta\gamma} \end{pmatrix}$$

Notice the assumption that $\alpha\delta \neq \beta\gamma$. This is typical behaviour, and usually what you want.

It's easy to get at parts of an expression.

Recall $A^{-1} = \begin{pmatrix} \frac{\delta}{\alpha\delta-\beta\gamma} & -\frac{\beta}{\alpha\delta-\beta\gamma} \\ -\frac{\gamma}{\alpha\delta-\beta\gamma} & \frac{\alpha}{\alpha\delta-\beta\gamma} \end{pmatrix}$

```
denominator(Ainverse[0,1])
```

evaluate

$\alpha\delta - \beta\gamma$

You can treat the matrix as a function

```
Ainverse(alpha=1,gamma=2)
```

evaluate

$$\begin{pmatrix} -\frac{\delta}{2\,\beta-\delta} & \frac{\beta}{2\,\beta-\delta} \\ \frac{2}{2\,\beta-\delta} & -\frac{1}{2\,\beta-\delta} \end{pmatrix}$$

This will work with any `SageMath` expression.

Rank

Minimum of number linearly independent rows, number of linearly independent columns

Recall from earlier,

```
C
```

evaluate

$$\begin{pmatrix} \alpha x_{11} + \beta x_{21} & \alpha x_{12} + \beta x_{22} & \alpha x_{13} + \beta x_{23} \\ \gamma x_{11} + \delta x_{21} & \gamma x_{12} + \delta x_{22} & \gamma x_{13} + \delta x_{23} \end{pmatrix}$$

```
rank(C)
```

evaluate

2

Rank of a product is the minimum of ranks

$$\mathbf{C} = \begin{pmatrix} \alpha x_{11} + \beta x_{21} & \alpha x_{12} + \beta x_{22} & \alpha x_{13} + \beta x_{23} \\ \gamma x_{11} + \delta x_{21} & \gamma x_{12} + \delta x_{22} & \gamma x_{13} + \delta x_{23} \end{pmatrix}$$

The matrix $\mathbf{C}^\top \mathbf{C}$ is awful to look at, but since the rank of a product is the minimum of the rank of the matrices being multiplied, its rank must be two (with `SageMath`'s usual optimistic assumptions).

```
rank( C.transpose()*C )
```

evaluate

2

Eigenvalues

Recall $\mathbf{A} = \begin{pmatrix} \alpha & \beta \\ \gamma & \delta \end{pmatrix}$

```
A.eigenvalues() # Returns a list
```

evaluate

$$\left[\frac{1}{2}\,\alpha + \frac{1}{2}\,\delta - \frac{1}{2}\,\sqrt{\alpha^2 - 2\,\alpha\delta + \delta^2 + 4\,\beta\gamma}, \frac{1}{2}\,\alpha + \frac{1}{2}\,\delta + \frac{1}{2}\,\sqrt{\alpha^2 - 2\,\alpha\delta + \delta^2 + 4\,\beta\gamma}\right]$$

- The eigenvalues of a real symmetric matrix are real.
- The expression under the square root sign will be non-negative if **A** is symmetric — that is, if $\beta = \gamma$.
- Sage doesn't care about this; imaginary numbers are fine.

Sum of eigenvalues equals the trace

For real symmetric matrices

```
eigenA = A.eigenvalues()
eigenA
```

evaluate

$$\left[\frac{1}{2}\,\alpha + \frac{1}{2}\,\delta - \frac{1}{2}\,\sqrt{\alpha^2 - 2\,\alpha\delta + \delta^2 + 4\,\beta\gamma}, \frac{1}{2}\,\alpha + \frac{1}{2}\,\delta + \frac{1}{2}\,\sqrt{\alpha^2 - 2\,\alpha\delta + \delta^2 + 4\,\beta\gamma}\right]$$

```
sum(eigenA)
```

evaluate

$\alpha + \delta$

The story continues. This is just a taste.

- Sage can save a lot of effort.
- It's great for working out examples that might be too painful otherwise.
- There are functions for many areas of math.
- If you know the math, looking at the documentation and getting `SageMath` to do it is pretty easy.

Copyright Information